Robinson’s (1937) definition of ordinal

Johan G. F. Belinfante
2008 February 28

```
In[1]:= SetDirectory["l:"]; << goedel.08feb27a; << tools.m

:Package Title: goedel.08feb27a                    2008 February 27 at 5:25 p.m.

It is now:  2008 Feb 28 at 15:1

Loading Simplification Rules

TOOLS.M                                            Revised 2008 February 12

weightlimit = 40
```

summary

In the **GOEDEL** program, the class of ordinals is defined, following J. R. Isbell, as follows. (A class **x** is full if each of its members is a subset, that is, if **subclass[U[x], x]**. The class **FULL** is the class of all full sets.)

```
In[2]:= class[x, subclass[intersection[FULL, P[x]], succ[x]]]

Out[2]= OMEGA
```

In other words, a set is an **ordinal** if every proper full subset is a member.

```
In[3]:= class[x, forall[y, implies[and[full[y], subclass[y, x]], or[member[y, x], equal[x, y]]]]]

Out[3]= OMEGA
```

Reference:

```
In[4]:= "J. R. Isbell, A Definition of Ordinal Numbers,
         American Mathematical Monthly, vol. 67 (1960), pp. 51-52.";
```

In addition to being concise, Isbell's definition does not require one to assume the axiom of regularity. If one does assume the axiom of regularity, one can alternatively characterize ordinals as those full sets whose members are also full.

```
In[5]:= "J. D. Monk, Introduction to Set Theory, McGraw-Hill Book Co., New York, 1969.";
```

A similar result also holds without the axiom of regularity:

```
In[6]:= intersection[REGULAR, FULL, P[FULL]]

Out[6]= OMEGA
```

R. M. Robinson's definition of an ordinal as a full, regular trichotomous set yields yet another expression for the class of ordinals:

```
In[7]:= class[x, and[member[x, REGULAR],
          subclass[cart[x, x], union[E, composite[inverse[E], SUCC]]], subclass[U[x], x]]]

Out[7]= intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]]
```

Reference:

```
In[8]:=  "R. M. Robinson, The theory of classes. A modification
          of von Neumann's system. Journal of Symbolic Logic, vol. 2
          (1937), pp. 29-36.  Reviewed by Paul Bernays, ibid., page 168.";
```

The equivalence of Robinson's definition is established in this notebook by showing that every member of an ordinal in Robinson's sense is full.

derivation

Lemma.

```
In[9]:= member[y, setpart[succ[x]]] // AssertTest

Out[9]= member[y, setpart[succ[x]]] ==
         or[and[equal[x, y], member[x, V]], and[member[x, V], member[y, x]]]

In[10]:= member[y_, setpart[succ[x_]]] :=
          or[and[equal[x, y], member[x, V]], and[member[x, V], member[y, x]]]
```

Theorem. Trichotomy.

```
In[11]:= SubstTest[implies, and[member[r, s], subclass[s, t]], member[r, t],
           {r → pair[x, y], s → cartsq[z], t -> union[E, composite[inverse[E], SUCC]]}] // Reverse

Out[11]= or[equal[x, y], member[x, y], member[y, x], not[member[x, z]], not[member[y, z]],
           not[subclass[cart[z, z], union[E, composite[inverse[E], SUCC]]]]] == True

In[12]:= or[equal[x_, y_], member[x_, y_],
           member[y_, x_], not[member[x_, z_]], not[member[y_, z_]],
           not[subclass[cart[z_, z_], union[E, composite[inverse[E], SUCC]]]]] := True
```

Lemma.

```
In[13]:= Map[not, SubstTest[and, implies[and[p4, p5, p6], or[p7, p8, p9]],
            implies[and[p1, p2, p3], not[p8]], implies[and[p1, p2, p3], not[p9]],
            not[implies[and[p1, p2, p3, p4, p5, p6], p7]],
            {p1 → member[w, REGULAR], p2 → member[u, v], p3 → member[v, w], p4 → member[w, x],
             p5 -> subclass[cartsq[x], union[Id, E, inverse[E]]], p6 → member[u, x],
             p7 → member[u, w], p8 -> equal[u, w], p9 → member[w, u]}]] // Reverse

Out[13]= or[member[u, w], not[member[u, v]], not[member[u, x]],
          not[member[v, w]], not[member[w, REGULAR]], not[member[w, x]],
          not[subclass[cart[x, x], union[E, composite[inverse[E], SUCC]]]]] == True

In[14]:= (% /. {u → u_, v → v_, w → w_, x → x_}) /. Equal → SetDelayed
```

Lemma.

```
In[15]:= Map[not, SubstTest[and, implies[and[p1, p4], p0], implies[and[p3, p4, p5], p7],
            implies[and[p2, p5, p7], p8], not[implies[and[p1, p2, p3, p4, p5, p6], p9]],
            {p0 → member[w, REGULAR], p1 → member[x, REGULAR], p2 → member[u, v], p3 → member[v, w],
             p4 → member[w, x], p5 → full[x], p6 -> subclass[cartsq[x], union[Id, E, inverse[E]]],
             p7 → member[v, x], p8 → member[u, x], p9 → member[u, w]}]] // Reverse

Out[15]= or[member[u, w], not[member[u, v]],
          not[member[v, w]], not[member[w, x]], not[member[x, REGULAR]],
          not[subclass[cart[x, x], union[E, composite[inverse[E], SUCC]]]],
          not[subclass[U[x], x]]] == True

In[16]:= (% /. {u → u_, v → v_, w → w_, x → x_}) /. Equal → SetDelayed
```

Eliminating all four variables yields:

```
In[17]:= Map[empty[composite[id[cart[V, V]], complement[#], id[cart[V, V]]]] &,
          SubstTest[class, pair[pair[u, v], pair[w, x]],
           or[member[u, w], not[member[u, v]], not[member[v, w]], not[member[w, x]],
            not[member[x, t]]], t → intersection[FULL, REGULAR, chains[union[Id, E]]]]]

Out[17]= subclass[U[intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]]], FULL] ==
          True

In[18]:= % /. Equal → SetDelayed
```

Lemma.

```
In[19]:= SubstTest[subclass, t, intersection[u, v, w],
           {t -> intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]],
            u → FULL, v → P[FULL], w → REGULAR}] // Reverse

Out[19]= subclass[intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]], OMEGA] ==
          True

In[20]:= % /. Equal → SetDelayed
```

Main Theorem. Equivalence of Robinson's definition of ordinals.

```
In[21]:= SubstTest[and, subclass[u, v], subclass[v, u],
          {u -> intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]], v -> OMEGA}]

Out[21]= equal[OMEGA, intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]]] == True

In[22]:= intersection[FULL, REGULAR, chains[composite[inverse[E], SUCC]]] := OMEGA
```

additional remarks

In this section, two further characterization of the class **OMEGA** are given which also do not depend on the axiom of regularity. Recall that the class **REGULAR** is defined in the **GOEDEL** program as follows:

```
In[23]:= class[x, forall[y, implies[subclass[y, image[E, y]], not[member[x, y]]]]]

Out[23]= REGULAR
```

The remarkable class **REGULAR** is its own power class, and its own sum class. It is a subclass of the class **FUND** of sets that are well-ordered by the membership relation:

```
In[24]:= class[x, WELLFOUNDED[composite[id[x], E]]]

Out[24]= FUND
```

Regularity implies well-foundedness.

```
In[25]:= implies[member[x, REGULAR], WELLFOUNDED[composite[id[x], E]]]

Out[25]= True
```

The converse holds only if one assumes the axiom of regularity. If **AxReg** is assumed, then **FUND = REGULAR = V**.

```
In[26]:= equal[FUND, REGULAR]

Out[26]= AxReg
```

In the presence of fullness, however, the classes **REGULAR** and **FUND** are interchangeable:

```
In[27]:= intersection[FUND, FULL]

Out[27]= intersection[FULL, REGULAR]
```

In view of this, one can also characterize the class **OMEGA** of ordinals in the following two ways, substituting **FUND** for **REGULAR**.

```
In[28]:= intersection[FUND, FULL, P[FULL]]

Out[28]= OMEGA

In[29]:= intersection[FUND, FULL, chains[union[Id, E]]]

Out[29]= OMEGA
```

All five characterizations of ordinals presented in this notebook do not depend on whether or not one assumes the axiom of regularity.